

» `cloudscale_server`

Provides a `cloudscale.ch` Server resource. This can be used to create, modify, and delete servers.

» Example Usage

```
# Create a new Server
resource "cloudscale_server" "web-worker01" {
  name           = "web-worker01"
  flavor_slug    = "flex-4"
  image_slug     = "debian-9"
  volume_size_gb = 10
  ssh_keys      = ["ssh-ed25519 AAAAC3NzaC1lZDI1NTE5AAAAIL2jzg1a23DfRVLQr3KT20QQYovqC"]
}
```

» Argument Reference

The following arguments are supported when creating new servers:

- **name** - (Required) Name of the new server. The name has to be a valid host name or a fully qualified domain name (FQDN).
- **flavor_slug** - (Required) The slug (name) of the flavor to use for the new server. Possible values can be found in our API documentation.
Note: If you want to update this value after initial creation, you must set **allow_stopping_for_update** to **true**.
- **image_slug** - (Required) The slug (name) of the image to use for the new server. Possible values can be found in our API documentation.
- **ssh_keys** - (Optional) A list of SSH public keys. Use the full content of your *.pub file here.
- **password** - (Optional) The password of the default user of the new server. When omitted, no password will be set.
- **zone_slug** - (Optional) You can specify a zone slug. Options include **lpg1** and **rma1**.
- **volume_size_gb** - (Optional) The size in GB of the SSD root volume of the new server. If this parameter is not specified, the value will be set to 10. The minimum value is 10.
- **bulk_volume_size_gb** - (Optional, Deprecated) The size in GB of the bulk storage volume of the new server. If this parameter is not specified, no bulk storage volume will be attached to the server. Valid values are multiples of 100.
- **use_public_network** - (Optional) Attach the public network interface to the new server. Can be **true** (default) or **false**. Use **interfaces** option for advanced setups.

- **use_private_network** - (Optional) Attach the **default** private network interface to the new server. Can be **true** or **false** (default). Use **interfaces** option for advanced setups.
- **use_ipv6** - (Optional) Enable/disable IPv6 on the public network interface of the new server. Can be **true** (default) or **false**.
- **interfaces** - (Optional) A list of interface configuration objects (see example). Each interface object has the following attributes:
 - **type** - (Required) The type of the interface. Can be **public** or **private**.
 - **network_uuid** (Required for **private** interfaces) The UUID of the private network this interface should be attached to.
- **user_data** - (Optional) User data (custom cloud-config settings) to use for the new server. Needs to be valid YAML. A default configuration will be used if this parameter is not specified or set to null. Use only if you are an advanced user with knowledge of cloud-config and cloud-init.
- **status** - (Optional) The desired state of a server. Can be **running** (default) or **stopped**.
- **allow_stopping_for_update** - (Optional) If true, allows Terraform to stop the instance to update its properties. If you try to update a property that requires stopping the instance without setting this field, the update will fail.

The following arguments are supported when updating servers:

- **name** - Name of the new server. The name has to be a valid host name or a fully qualified domain name (FQDN).
- **volume_size_gb** - The size in GB of the SSD root volume of the new server.
- **interfaces** - A list of interface configuration objects. Each interface object has the following attributes:
 - **type** - (Required) The type of the interface. Can be **public** or **private**.
 - **network_uuid** (Required for **private** interfaces) The UUID of the private network this interface should be attached to.
- **status** - The desired state of a server. Can be **running** (default) or **stopped**.

» Attributes Reference

In addition to the arguments listed above, the following computed attributes are exported:

- **id** - The UUID of this server.
- **href** - The cloudscale.ch API URL of the current resource.
- **ssh_fingerprints** - A list of SSH host key fingerprints (strings) of this server.

- **ssh_host_keys** - A list of SSH host keys (strings) of this server.
- **volumes** - A list of volume objects attached to this server. Each volume object has three attributes:
 - **device_path** - The path (string) to the volume on your server (e.g. `/dev/vda`)
 - **size_gb** - The size (int) of the volume in GB. Typically matches **volume_size_gb** or **bulk_volume_size_gb**.
 - **type** - A string. Either **ssd** or **bulk**.
- **public_ipv4** - The first **public** IPv4 address of this server. The returned IP address may be "" if the server does not have a public IPv4.
- **private_ipv4** - The first **private** IPv4 address of this server. The returned IP address may be "" if the server does not have private networking enabled.
- **public_ipv6** - The first **public** IPv6 address of this server. The returned IP address may be "" if the server does not have a public IPv6.
- **interfaces** - A list of interface objects attached to this server. Each interface object has the following attributes:
 - **network_uuid** - The UUID of the network the interface is attached to.
 - **network_name** - The name of the network the interface is attached to.
 - **network_href** - The `cloudscale.ch` API URL of the network the interface is attached to.
 - **type** - Either **public** or **private**. Public interfaces are connected to the Internet, while private interfaces are not.
 - **addresses** - A list of address objects:
 - * **address** - An IPv4 or IPv6 address that has been assigned to this server.
 - * **gateway** - An IPv4 or IPv6 address that represents the default gateway for this interface.
 - * **prefix_length** - The prefix length for this IP address, typically 24 for IPv4 and 128 for IPv6.
 - * **reverse_ptr** - The PTR record (reverse DNS pointer) for this IP address. If you use an FQDN as your server name it will automatically be used here.
 - * **version** - The IP version, either 4 or 6.

» **cloudscale_server_group**

Provides a `cloudscale.ch` Server Group resource. This can be used to create, and delete server groups.

» Example Usage

```
# Add a server group with anti affinity
resource "cloudscale_server_group" "web-worker-group" {
  name = "web-worker-group"
  type = "anti-affinity"
}

# Create three new servers in that group
resource "cloudscale_server" "web-worker01" {
  count          = 3
  name           = "web-worker${count.index}"
  flavor_slug    = "flex-4"
  image_slug     = "debian-9"
  server_group_ids = [cloudscale_server_group.web-worker-group.id]
  ssh_keys       = ["ssh-ed25519 AAAAC3NzaC1lZDI1NTE5AAAAIL2jzglA23DfRVLQr3KT20QQYovqCCN3"]
}
```

» Argument Reference

The following arguments are supported when creating server groups:

- **name** - (Required) Name of the new server group.
- **type** - (Required) The type of the server group can currently only be "anti-affinity".
- **zone_slug** - (Optional) You can specify a zone slug. Options include `lpg1` and `rma1`.

» Attributes Reference

In addition to the arguments listed above, the following computed attributes are exported:

- **href** - The cloudscale.ch API URL of the current resource.

» cloudscale__volume

Provides a cloudscale.ch Volume (block storage) resource. This can be used to create, modify, and delete Volumes.

» Example Usage

```
# Create a new Server
```

```

resource "cloudscale_server" "web-worker01" {
  name          = "web-worker01"
  flavor_slug   = "flex-4"
  image_slug    = "debian-9"
  ssh_keys     = ["ssh-ed25519 AAAAC3NzaC1lZDI1NTE5AAAAIL2jzglA23DfRVLQr3KT20QQYovqCCN3c1Hrjn"]
}

# Add a Volume to web-worker01
resource "cloudscale_volume" "web-worker01-volume" {
  name          = "web-worker-data"
  size_gb       = 100
  type          = "ssd"
  server_uuids = [cloudscale_server.web-worker01.id]
}

```

» Argument Reference

The following arguments are supported when creating/changing Volumes:

- **name** - (Required) Name of the new volume.
- **size_gb** - (Required) The volume size in GB. Valid values are multiples of 1 for type "ssd" and multiples of 100 for type "bulk".
- **zone_slug** - (Optional) You can specify a zone slug. Options include `lpg1` and `rma1`.
- **type** - (Optional) For SSD/NVMe volumes specify "ssd" (default) or use "bulk" for our HDD cluster with NVMe caching. This is the only attribute that cannot be altered.
- **server_uuids** - (Optional) A list of server UUIDs. Default to an empty list. Currently a volume can only be attached to one server UUID.

» Attributes Reference

In addition to the arguments listed above, the following computed attributes are exported:

- **href** - The cloudscale.ch API URL of the current resource.

» cloudscale__network

Provides a cloudscale.ch Private Network resource. This can be used to create, modify, and delete networks.

» Example Usage

```
# Create a new private network
resource "cloudscale_network" "privnet" {
  name      = "privnet"
  zone_slug = "lpg1"
  mtu       = "9000"
}

# Add a server with two interfaces:
# - one attached to the public network
# - one attached to the private network "privnet"
resource "cloudscale_server" "gw" {
  name      = "gateway"
  zone_slug = "lpg1"
  flavor_slug = "flex-4"
  image_slug = "debian-9"
  volume_size_gb = 20
  ssh_keys   = ["ssh-ed25519 AAAAC3NzaC1lZDI1NTE5AAAAIL2jz gla23DfRVLQr3KT20QQYovqC"]
  interfaces {
    type = "public"
  }
  interfaces {
    type      = "private"
    network_uuid = cloudscale_network.privnet.id
  }
}
```

» Argument Reference

The following arguments are supported when creating/changing Networks:

- **name** - (Required) Name of the new network.
- **zone_slug** - (Optional) You can specify a zone slug. Options include `lpg1` and `rma1`.
- **mtu** - (Optional) You can specify the MTU size for the network, defaults to 9000.

» Attributes Reference

In addition to the arguments listed above, the following computed attributes are exported:

- **href** - The cloudscale.ch API URL of the current network.

- **subnets** - A list of subnet objects that are used in this network. Each subnet object has the following attributes:
 - **cidr** - The CIDR notation of the subnet.
 - **href** - The cloudscale.ch API URL of this subnet.
 - **uuid** - The UUID of this subnet.

» cloudscale_floating_ip

Provides a cloudscale.ch Floating IP to represent a publicly-accessible static IP address or IP network that can be assigned to one of your cloudscale.ch servers. Floating IPs can be moved between servers. Possible use cases include: High-availability, non-disruptive maintenance, multiple IPs per server, or re-using the same IP after replacing a server.

» Example Usage

```
# Create a new Server
resource "cloudscale_server" "web-worker01" {
  name          = "web-worker01"
  flavor_slug   = "flex-4"
  image_slug    = "debian-9"
  ssh_keys     = ["ssh-ed25519 AAAAC3NzaC1lZDI1NTE5AAAAIL2jzglA23DfRVLQr3KT20QQYovqCCN3c1Hrj"]
}

# Add a Floating IPv4 address to web-worker01
resource "cloudscale_floating_ip" "web-worker01-vip" {
  server        = cloudscale_server.web-worker01.id
  ip_version    = 4
  reverse_ptr   = "vip.web-worker01.example.com"
}

# Add a Floating IPv6 network to web-worker01
resource "cloudscale_floating_ip" "web-worker01-net" {
  server        = cloudscale_server.web-worker01.id
  ip_version    = 6
  prefix_length = 56
}
```

» Argument Reference

The following arguments are supported when adding Floating IPs:

- **server** - (Required) Assign the Floating IP to this server (UUID).

- **ip_version** - (Required) 4 or 6, for an IPv4 or IPv6 address or network respectively.
- **prefix_length** - (Optional) If you want to assign an entire network instead of a single IP address to your server, you must specify the prefix length. Currently, there is only support for **ip_version=6** and **prefix_length=56**.
- **region_slug** - (Optional) You can specify a region slug. Options include **lpg** and **rma**.
- **reverse_ptr** - (Optional) You can specify the PTR record (reverse DNS pointer) in case of a single Floating IP address.

The following arguments are supported when updating Floating IPs:

- **server** - (Required) (Re-)Assign the Floating IP to this server (UUID).

» Attributes Reference

In addition to the arguments listed above, the following computed attributes are exported:

- **href** - The `cloudscale.ch` API URL of the current resource.
- **network** - The CIDR notation of the Floating IP address or network, e.g. `192.0.2.123/32`.
- **next_hop** - The IP address of the server that your Floating IP is currently assigned to.